
Inside the Mouse�

The Program�

'{$STAMP BS2} 
'  Mouse code for BS2 and PAK-VI 
 
pakout con 15 
pakin con 14 
baud con 84 
fpin con 13 
 
keyin var byte 
' This assumes a 3 byte packet 
' Some mice can do 4 byte packets 
' (mainly those with wheels) but 
' you have to turn on that mode 
p1 var byte   ' mouse status 
xreg var byte ' raw x motion 
xmov var word ' x accumulator 
yreg var byte ' raw y motion 
ymov var word ' y accumulator 
 
' flags in p1 
yover var p1.bit7   ' y overflow 



xover var p1.bit6   ' x overflow 
ysign var p1.bit5   ' y sign 
xsign var p1.bit4   ' x sign 
midbtn var p1.bit2  ' button flags 
rightbtn var p1.bit1 
leftbtn var p1.bit0 
 
high fpin 
 
' Reset PAK and wait for device reset 
serout pakout,baud,[$FF] 
pause 500 
xmov=0 
ymov=0 
' Enter raw mode 
serout pakout,baud,[2] 
' Set remote mode 
serout pakout,baud,[$0B,$F0] 
serin pakin\fpin,baud,[keyin] 
' should get $FA 
if keyin=$FA then top 
Debug "Didn't get $FA from remote mode set",cr 
 
top: 
' read movement packet 
serout pakout,baud,[$0B,$EB] 
serin pakin\fpin,baud,[keyin,p1,xreg,yreg] 
if keyin=$FA then parse 
debug "Didn't get $FA on mouse poll",cr 
goto top 
 
parse: 
if yover or xover then overflow 
if xsign=0 then xplus 
' convert negative number 
xmov = xmov - ((xreg-1)^$FF) 
goto doy 
xplus: 
xmov=xmov+xreg 
doy: 
if ysign=0 then yplus 
' convert negative number 
ymov=ymov- ((yreg-1)^$FF) 
goto ydone 
yplus: 
ymov=ymov+yreg 
ydone: 
 
' Display results 
debug cls, sdec xmov, " ", sdec ymov ,cr 
if leftbtn=0 then checkmid 
Debug "Left button down",cr 
checkmid: 
if midbtn=0 then checkr 
Debug "Mid button down",cr 
checkr: 
if rightbtn=0 then top 
Debug "Right button down",cr 
goto top 
 



overflow: 
Debug "Overflow occured",cr 
goto top 


